
Laboratorio Fond. Ricerca Operativa Prof. E. Amaldi

Mix Produttivo

Un’azienda vuole pianificare il livello di produzione di 3 prodotti (A1,A2,A3). Vi sono
in totale 22 giorni di produzione disponibili in un mese. Nella seguente tabella sono
indicati: la domanda massima (in quintali), il prezzo di vendita (in dollari/quintale), il
costo di produzione per quintale di prodotto e la quota di produzione (cioè il massimo
numero di quintali prodotti in 1 giorno se tutte le risorse a disposizione fossero usate per
un solo tipo di prodotto).

Prodotto A1 A2 A3

Domanda massima 5300 4500 5400
Prezzo di vendita $124 $109 $115

Costo di produzione $73.30 $52.90 $65.40
Quota di produzione 500 450 550

• Formulare un modello AMPL per determinare il piano di produzione che massimizza
il guadagno totale.

• Come cambiano il modello matematico, il modello e i dati AMPL e la soluzione,
se per ogni prodotto introduciamo un costo fisso di attivazione della linea di pro-
duzione:

Prodotto A1 A2 A3

Costo di attivazione $170000 $150000 $100000

• Come cambiano il modello matematico, il modello e i dati AMPL e la soluzione, se
oltre al costo fisso di attivazione aggiungiamo anche il vincolo che se una linea di
produzione viene attivata, è necessaria la produzione di un lotto minimo di prodotto:

Prodotto A1 A2 A3

Lotto minimo 20 20 16

Soluzione

Formulazione.

• Indici: sia i un indice sull’insieme {1, 2, 3}.

• Parametri:

– sia P il numero di giorni di produzione in un mese;

– sia di la domanda massima per il prodotto i;

– sia vi il prezzo di vendita per il prodotto i;

– sia ci il costo di produzione del prodotto i;

– sia qi la quota di produzione massima del prodotto i;

– sia ai il costo di attivazione dell’impianto per il prodotto i

[parametro necessario solo per le domande aggiuntive sui costi d’attivazione e
sui lotti minimi];

– sia li il lotto minimo di prodotto i

[parametro necessario solo per la domanda aggiuntiva sul lotto minimo].

• Variabili:

– sia xi la quantità di prodotto i. I limiti sulle variabili sono xi ≥ 0.

– sia yi = 1 se la linea di produzione per il prodotto i è attivata, e 0 se non lo è.
yi è una variabile binaria
[variabili necessarie solo per le domande aggiuntive sul costo di attivazione e
sul lotto minimo].

• Funzione obiettivo:
max

∑

i

((vi − ci)xi − aiyi)

• Vincoli:

1. (richiesta): per ogni i, xi ≤ di;

2. (produzione):
∑

i
xi

qi

≤ P ;

3. (attivazione): per ogni i, xi ≤ Pqiyi

[vincolo necessario solo per le domanda aggiuntive sul costo di attivazione e sul
lotto minimo];

4. (lotto minimo): per ogni i, xi ≥ liyi

[vincolo necessario solo per la domanda aggiuntiva sul lotto minimo].

Modello di AMPL

# mixproduttivo.mod - file modello per AMPL - mix produttivo con vincoli agg.

set PRODOTTI;

param giorni >= 0;

param domanda { PRODOTTI } >= 0;

param vendita { PRODOTTI } >= 0;

param costo { PRODOTTI } >= 0;

param quota { PRODOTTI } >= 0;

param costo_attiv { PRODOTTI } >= 0; # costi di attivazione

param lottomin { PRODOTTI } >= 0; # lotti minimi

var x { PRODOTTI } >= 0; # quantita‘ di prodotto

var y { PRODOTTI } >= 0, binary; # attivazione linee produzione

maximize guadagno: sum {i in PRODOTTI}

((vendita[i] - costo[i]) * x[i] - costo_attiv[i] * y[i]);

subject to richiesta {i in PRODOTTI}:

x[i] <= domanda[i];

subject to produzione:

sum {i in PRODOTTI} (x[i] / quota[i]) <= giorni;

subject to attivazione {i in PRODOTTI}:

x[i] <= giorni * quota[i] * y[i];

subject to lotto_minimo {i in PRODOTTI}:

x[i] >= lottomin[i] * y[i];

Dati di AMPL

# mixprod.dat - file di dati per AMPL - mix produttivo

set PRODOTTI := A1 A2 A3 ;

param giorni := 22;

param : domanda vendita costo quota costo_attiv lottomin :=

A1 5300 124 73.30 500 170000 20

A2 4500 109 52.90 450 150000 20

A3 4400 115 65.40 550 100000 16 ;

Soluzione di CPLEX.

CPLEX 7.1.0: optimal integer solution; objective 220690

5 MIP simplex iterations

0 branch-and-bound nodes

ampl: display x;

x [*] :=

A1 0

A2 4500

A3 4400

;

ampl: display y;

y [*] :=

A1 0

A2 1

A3 1

;

Documento preparato da: Leo Liberti 4


